REFERENCES

Cranfield , R.R., and Geldart, D. (1974). Large particle fluidization. Chemical Engineering Science, 29, 935-947.

Davidson, J.F., and Harrison, D. (1963). Fluidized Particles. Cambridge: Cambridge University Press.

Harrison, D., and Davidson, J.F. (1971). Fluidization. London: Academic Press.

Kunii, D., and Levenspiel, O. (1977). Fluidization Engineering. New York: Robert E. Krieger.

Maneri, C.C., and Mendelson, H.D. (1968). The rise velocity of bubbles in tubes and rectangular channels as predicted by wave theory. AIChE Journal, 14(2), 295-300.

Nguyen, X.T., and Leung, L.S. (1972). A note on bubble formation at an orifice in a fluidized bed. Chemical Engineering Science, 27, 1748-1750.

Wilkes, J.O. (1999). Fluid Mechanics for Chemical Engineers. New Jersey: Prentice Hall.

Yates, J.G., Rowe, P.N., and Cheesman, D.J. (1984). Gas Entry Effects in Fluidized Bed Reactors. AIChE Joural, 30(6), 890-894.

Zenz, F.A. (1968). Bubble formation and grid design. Institutution of Chemical Engineering Symposium, 30, 136-139.

APPENDICS

APPENDIX A Finite element method

The finite element is the numerical method for calculating the approximate values of the mathematical problems. By dividing the area of the problem into the elements and creating the equation of each element corresponding to the partial differential equation. The elemental equations were summarized to the system of equations for setting the boundary conditions. Gauss elimination method was used to solve the approximate values from the system of equations. These phenomena of gas streamlines assisted the operators to design the distributor in the fluidized bed, since many unique properties of fluidized beds were related to the presence of bubbles.

The problems of gas streamlines in the fluidized bed were solved by the two-dimensional finite element method. These problems of gas streamlines were derived in term of the partial differential equation.

$$\frac{\partial}{\partial x}\left(k\frac{\partial \bar{\psi}}{\partial x}\right)+\frac{\partial}{\partial y}\left(k\frac{\partial \bar{\psi}}{\partial y}\right) = 0 \qquad \text{(A1)}$$

If the exact value of gas streamlines, $\bar{\psi}$, was known and substituted into the left-hand-side of Equation A1, the result in the right-hand-side was equal to zero. Since the exact value was not solved from the partial differential equation, the approximate value, ψ, substituted into the left-hand-side of Equation A1 resulted in the residue, R, in the partial differential equation

$$\frac{\partial}{\partial x}\left(k\frac{\partial \psi}{\partial x}\right)+\frac{\partial}{\partial y}\left(k\frac{\partial \psi}{\partial y}\right) = R \qquad \text{(A2)}$$

In the Galerkin method, the finite element equations were created by multiplying the residue with the weighting function, W, and integrating around the area of the element to be zero.

$$\int_{\Omega} W_i R \, d\Omega = 0 \tag{A3}$$

Because triangular element consisted of three nodes, therefore, it was necessary to use three equations for the solution. By substituting Equation A2 into Equation A3:

$$\int_{\Omega} W_i \left(\frac{\partial}{\partial x}\left(k \frac{\partial \psi}{\partial x} \right) + \frac{\partial}{\partial y}\left(k \frac{\partial \psi}{\partial y} \right) \right) d\Omega = 0 \tag{A4}$$

and integrating Equation A4 by Gauss's theorem:

$$\int_{\Omega} u \left(\nabla \cdot \vec{V} \right) d\Omega = \int_{\Gamma} u \left(\vec{V} \cdot \hat{n} \right) d\Gamma - \int_{\Omega} \left(\nabla u \cdot \vec{V} \right) d\Omega \tag{A5a}$$

where

$$u_i = W_i \tag{A5b}$$

$$\nabla = \frac{\partial}{\partial x} \hat{i} + \frac{\partial}{\partial y} \hat{j} \tag{A5c}$$

$$\vec{V} = k \frac{\partial \psi}{\partial x} \hat{i} + k \frac{\partial \psi}{\partial y} \hat{j} \tag{A5d}$$

$$\hat{n} = n_x \hat{i} + n_y \hat{j} \tag{A5e}$$

$$\vec{V} \cdot \hat{n} = k \frac{\partial \psi}{\partial x} n_x + k \frac{\partial \psi}{\partial y} n_y \tag{A5f}$$

$$u \left(\vec{V} \cdot \hat{n} \right) = W_i \left(k \frac{\partial \psi}{\partial x} n_x + k \frac{\partial \psi}{\partial y} n_y \right) \tag{A5g}$$

$$\nabla u = \frac{\partial W_i}{\partial x} \hat{i} + \frac{\partial W_i}{\partial y} \hat{j} \tag{A5h}$$

$$\nabla u \cdot \vec{V} \quad = \quad \frac{\partial W_i}{\partial x} k \frac{\partial \Psi}{\partial x} + \frac{\partial W_i}{\partial y} k \frac{\partial \Psi}{\partial y} \tag{A5i}$$

Substituting Equation A5f-I into Equation A4 and assuming that $W_i = N_i$:

$$\int_{\Gamma} N_i \left(k \frac{\partial \psi}{\partial x} n_x + k \frac{\partial \psi}{\partial y} n_y \right) d\Gamma - \iint_{\Omega} \left(\frac{\partial N_i}{\partial x} k \frac{\partial \Psi}{\partial x} + \frac{\partial N_i}{\partial y} k \frac{\partial \Psi}{\partial y} \right) d\Omega \qquad i = 1,2,3 \tag{A6}$$

$$\iint_{\Omega} \left(\frac{\partial N_i}{\partial x} k \frac{\partial \Psi}{\partial x} + \frac{\partial N_i}{\partial y} k \frac{\partial \Psi}{\partial y} \right) d\Omega \quad = \quad \int_{\Gamma} N_i \left(k \frac{\partial \psi}{\partial x} n_x + k \frac{\partial \psi}{\partial y} n_y \right) d\Gamma \qquad i = 1,2,3 \tag{A7}$$

Since there were three equations in the element, the finite element equations were written in term of matrix.

$$\iint_{\Omega} \left(\left\{ \frac{\partial N}{\partial x} \right\} k \frac{\partial \Psi}{\partial x} + \left\{ \frac{\partial N}{\partial y} \right\} k \frac{\partial \Psi}{\partial y} \right) d\Omega \quad = \quad \int_{\Gamma} \{N\} \left(k \frac{\partial \psi}{\partial x} n_x + k \frac{\partial \psi}{\partial y} n_y \right) d\Gamma \tag{A8}$$

In each element, the distribution of gas streamlines in triangular element was assumed that

$$\psi \quad = \quad \psi(x_i, y_i) \quad = \quad \alpha_1 + \alpha_2 x_i + \alpha_3 y_i \qquad i = 1,2,3 \tag{A9a}$$

or

$$\psi(x, y) \quad = \quad [N_1(x,y) \quad N_2(x,y) \quad N_3(x,y)] \begin{Bmatrix} \psi_1 \\ \psi_2 \\ \psi_3 \end{Bmatrix} \tag{A9b}$$

$$N_i \quad = \quad \frac{1}{2A} (a_i + b_i x + c_i y) \qquad i = 1,2,3 \tag{A9c}$$

A = Area of triangular element

$$A \quad = \quad \frac{1}{2} [x_2 (y_3 - y_1) + x_1 (y_2 - y_3) + x_3 (y_1 - y_2)] \tag{A9d}$$

$$\psi = \psi(x, y) = \underset{(1\times3)}{[N]} \underset{(3\times1)}{\{\psi\}} \tag{A9e}$$

$$\frac{\partial \psi}{\partial x} = \underset{(1\times3)}{\left[\frac{\partial N}{\partial x}\right]} \underset{(3\times1)}{\{\psi\}} \tag{A9f}$$

$$\frac{\partial \psi}{\partial y} = \underset{(1\times3)}{\left[\frac{\partial N}{\partial y}\right]} \underset{(3\times1)}{\{\psi\}} \tag{A9g}$$

$$\frac{\partial N_i}{\partial x} = \frac{b_i}{2A} \tag{A9h}$$

$$\frac{\partial N_i}{\partial y} = \frac{c_i}{2A} \tag{A9i}$$

Substituting Equation A9f and A9g into Equation A8, the finite element equations became:

$$\int_{\Omega} \left(\underset{(3\times1)}{\left\{\frac{\partial N}{\partial x}\right\}} k \underset{(1\times3)}{\left[\frac{\partial N}{\partial x}\right]} + \underset{(3\times1)}{\left\{\frac{\partial N}{\partial y}\right\}} k \underset{(1\times3)}{\left[\frac{\partial N}{\partial y}\right]} \right) d\Omega \underset{(3\times1)}{\{\psi\}} = \int_{\Gamma} \underset{(3\times1)}{\{N\}} \left(k \frac{\partial \psi}{\partial x} n_x + k \frac{\partial \psi}{\partial y} n_y \right) d\Gamma$$

(A10)

From Equation A10, the finite element equations were written in the simple term of:

$$\underset{(3\times3)}{[K]_e} \underset{(3\times1)}{\{\psi\}_e} = \underset{(3\times1)}{\{F\}_e} \tag{A11}$$

where subscript, e, was element matrix.

Substituting Equation A9h and A9i into $\underset{(3\times3)}{[K]_e}$ in Equation A11.

$$K_{ij} = \int_A k\left(\frac{\partial N_i}{\partial x}\frac{\partial N_j}{\partial x} + \frac{\partial N_i}{\partial y}\frac{\partial N_j}{\partial y}\right)dx\,dy \quad i,j = 1,2,3 \tag{A12}$$

$$K_{ij} = k\int_A\left(\frac{b_i}{2A}\frac{b_j}{2A} + \frac{c_i}{2A}\frac{c_j}{2A}\right)dx\,dy \tag{A13}$$

$$K_{ij} = \frac{k}{4A^2}\left(b_i b_j + c_i c_j\right)\int_A dx\,dy \tag{A14}$$

$$K_{ij} = \frac{k}{4A}\left(b_i b_j + c_i c_j\right) \quad i,j = 1,2,3 \tag{A15}$$

or

$$\underset{(3\times3)}{[K]} = kA\underset{(3\times2)}{[B]^T}\underset{(2\times3)}{[B]} \tag{A16}$$

$$[B] = \frac{1}{2A}\begin{bmatrix} b_1 & b_2 & b_3 \\ c_1 & c_2 & c_3 \end{bmatrix} \tag{A17}$$

After all of elemental matrices were created, they were combined to the system of equations:

$$\sum(\text{element equations}) \Rightarrow [K]_{sys}\{\psi\}_{sys} = \{F\}_{sys} \tag{A18}$$

where subscript, sys, is the system of equations.

APPENDIX B Gauss elimination method

Gauss elimination method was used to solve the values of gas streamlines in the system of equations. The solution consisted of 'n' equations and 'n' variables, was displayed in Equation B1a-n.

$$a_{11} x_1 + a_{12} x_2 + a_{13} x_3 + \cdots\cdots + a_{1n} x_n = b_1 \qquad \text{(B1a)}$$

$$a_{21} x_1 + a_{22} x_2 + a_{23} x_3 + \cdots\cdots + a_{2n} x_n = b_2 \qquad \text{(B1b)}$$

$$a_{31} x_1 + a_{32} x_2 + a_{33} x_3 + \cdots\cdots + a_{3n} x_n = b_3 \qquad \text{(B1c)}$$

$$\vdots \qquad \vdots \qquad \vdots \qquad \qquad \vdots \qquad \vdots$$

$$a_{n1} x_1 + a_{n2} x_2 + a_{n3} x_3 + \cdots\cdots + a_{nn} x_n = b_n \qquad \text{(B1n)}$$

By using the forward elimination, Equation B1a was divided by the coefficient of Equation B1a, a_{11}, and multiplied with the coefficient of Equation B1b, a_{21}.

$$x_1 + \frac{a_{12}}{a_{11}} x_2 + \frac{a_{13}}{a_{11}} x_3 + \cdots\cdots + \frac{a_{1n}}{a_{11}} x_n = \frac{b_1}{a_{11}} \qquad \text{(B2)}$$

$$a_{21} x_1 + a_{21}\frac{a_{12}}{a_{11}} x_2 + a_{21}\frac{a_{13}}{a_{11}} x_3 + \cdots\cdots + a_{21}\frac{a_{1n}}{a_{11}} x_n = a_{21}\frac{b_1}{a_{11}} \qquad \text{(B3)}$$

Subtracting Equation B1a by Equation B3, Equation B1a became:

$$\left(a_{22} - a_{21}\frac{a_{12}}{a_{11}}\right)x_2 + \left(a_{23} - a_{21}\frac{a_{13}}{a_{11}}\right)x_3 + \cdots\cdots + \left(a_{2n} - a_{21}\frac{a_{1n}}{a_{11}}\right)x_n = b_2 - a_{21}\frac{b_1}{a_{11}}$$

(B4)

or

$$a'_{22} x_2 + a'_{23} x_3 + \cdots\cdots + a'_{2n} x_n = b'_2 \qquad \text{(B5)}$$

To repeat the step of the calculation from Equation B1a to Equation B1b, the system of equations was:

$$a_{11}x_1 + a_{12}x_2 + a_{13}x_3 + \cdots + a_{1n}x_n = b_1 \quad \text{(B6a)}$$

$$a'_{22}x_2 + a'_{23}x_3 + \cdots + a'_{2n}x_n = b'_2 \quad \text{(B6b)}$$

$$a'_{32}x_2 + a'_{33}x_3 + \cdots + a'_{3n}x_n = b'_3 \quad \text{(B6c)}$$

$$\vdots \qquad \vdots \qquad \vdots \qquad \vdots$$

$$a'_{n2}x_2 + a'_{n3}x_3 + \cdots + a'_{nn}x_n = b'_n \quad \text{(B6n)}$$

From the first round of elimination method, all values in the first column of Equation B6b-n except Equation B6a were equal to zero. The problem was solved by using forward elimination from the second round to the n-1 round of elimination method until the system of equations was proper to solve the results of streamlines by back substitution.

$$a_{11}x_1 + a_{12}x_2 + a_{13}x_3 + \cdots + a_{1n}x_n = b_1 \quad \text{(B7a)}$$

$$a'_{22}x_2 + a'_{23}x_3 + \cdots + a'_{2n}x_n = b'_2 \quad \text{(B7b)}$$

$$a''_{33}x_3 + \cdots + a''_{3n}x_n = b''_3 \quad \text{(B7c)}$$

$$\vdots \qquad \vdots \qquad \vdots$$

$$a^{(n-1)}_{nn}x_n = b^{(n-1)}_{nn} \quad \text{(B7n)}$$

From the system of Equation B7a-n, the result of x_n was calculated directly from Equation B7n:

$$x_n = \frac{b^{(n-1)}_n}{a^{(n-1)}_{nn}} \quad \text{(B8)}$$

and the results of $x_{n-1}, x_{n-2}, \ldots, x_2, x_1$ were computed by:

$$x_i = \frac{b^{(i-1)}_i - \sum_{j=i+1}^{n} a^{(i-1)}_{ij} x_j}{a^{(i-1)}_{ii}} \quad \text{(B9)}$$

APPENDIX C Finite element code

```
C       PROGRAM FINITE
        PARAMETER (MXNODE=300, MXELE=600)
        IMPLICIT REAL*8 (A-H,O-Z)
        DIMENSION COORD(MXNODE,2), SL(MXNODE), TEXT(20)
        DIMENSION SYSK(MXNODE,MXNODE), SYSF(MXNODE)
        DIMENSION PER(MXELE)
        INTEGER NNUM(MXELE,3), COND(MXNODE)
C       MXNODE  =   maximum number of nodes
C       MXELE   =   maximum number of elements
C       COORD   =   coordinates of nodes
C       SL      =   streamline
C       SYSK    =   [K]sys
C       SYSF    =   {F}sys
C       PER     =   permeability
C       NNUM    =   nodal numbers were combined to one element
C       COND    =   condition of calculation
C       If the users knew the gas streamline in this point, select          1.
C       If the users do not knew the gas streamline in this point, select   0.
        OPEN(7,FILE='INPUT.DAT')
C       To open 'input.dat' file to read the input data.
101     FORMAT(20A4)
        READ(7,101) TEXT
C       To read character.
        READ(7,*) NODE,NELEM
C       To read a number of nodes and elements.
        IF(NODE.GT.MXNODE) WRITE(6,102) NODE
C       To check a number of nodes.
```

```
102    FORMAT(/,' PLEASE INCREASE THE PARAMETER MXNODE
       TO' *    , I5)
       IF(NODE.GT.MXNODE) STOP
       IF(NELEM.GT.MXELE) WRITE(6,103) NELEM
C      To check a number of elements.
103    FORMAT(/,' PLEASE INCREASE THE PARAMETER MXELE TO'
     *  , I5)
       IF(NELEM.GT.MXELE) STOP
       READ(7,101) TEXT
C      To read character.
       DO 104 IP=1,NODE
       READ(7,*) I, COND(I), (COORD(I,K), K=1,2), SL(I)
C      To read nodal numbers, conditions, nodal coordinates and value of gas
C      streamlines.
       IF(I.NE.IP) WRITE(6,105) IP
C      To check the data of each node.
105    FORMAT(/, ' NODE NO. ',I5, ' IN DATA FILE IS MISSING')
       IF(I.NE.IP) STOP
104    CONTINUE
       READ(7,101) TEXT
 C     To read character.
       DO 106 IE=1,NELEM
       READ(7,*) I, (NNUM(I,J), J=1,3),PER(I)
C      To check element numbers, nodal numbers, permeability.
       IF(I.NE.IE) WRITE(6,107) IE
C      To check the data of each element.
107    FORMAT(/, ' ELEMENT NO. ', I5, ' IN DATA FILE IS MISSING')
       IF(I.NE.IE) STOP
106    CONTINUE
```

```
      WRITE(6,108)
108   FORMAT(' THE FINITE ELEMENT PROGRAM')
      NEQ=NODE
C     A number of equations in the solution equaled the number of nodes.
      DO 109 I=1,NEQ
      SYSF(I)=0.
C     To clear the values of {F}sys.
109   CONTINUE
      DO 110 I=1,NEQ
      DO 110 J=1,NEQ
      SYSK(I,J)=0.
C     To clear the values of [K]sys.
110   CONTINUE
      CALL ELEMENT(NELEM,NNUM,COORD,PER,SYSK,
     * MXNODE,MXELE)
C     To call subroutine for calculating elemental matrices.
      CALL BOUNDARY(NODE,COND,SL,SYSK,SYSF,MXNODE)
C     To call subroutine for setting the boundary conditions.
      CALL GAUSS(NEQ,SYSK,SYSF,SL,MXNODE)
C     To call subroutine for solving the results of gas streamlines.
      OPEN(8,FILE='Output.DAT')
C     To open 'output.dat' file to record the results of gas streamline.
      WRITE(8,111)
111   FORMAT('NODE',10X,'', 'X',10X, 'Y',10X, 'STREAMLINE')
      DO 112 IP=1,NODE
      WRITE(8,113) IP,COORD(IP,1),COORD(IP,2),SL(IP)
C     To print the nodal numbers, coordinates and streamlines.
113   FORMAT(I6, F10.5,F10.5,F10.5)
112   CONTINUE
```

```
      STOP
      END
C-------------------------------------------------------
      SUBROUTINE ELEMENT(NELEM,NNUM,COORD,PER,
     * SYSK,MXNODE,MXELE)
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION COORD(MXNODE,2), SYSK(MXNODE,MXNODE)
      DIMENSION AKC(3,3), B(2,3), BT(3,2)
      INTEGER NNUM(MXELE,3)
      DIMENSION PER(NELEM)
```

C $\quad B = \begin{bmatrix} b_1 & b_2 & b_3 \\ c_1 & c_2 & c_3 \end{bmatrix}$

C $\quad BT = \begin{bmatrix} b_1 & c_1 \\ b_2 & c_2 \\ b_3 & c_3 \end{bmatrix}$

C Area of the element

C $\quad \psi(x,y) = [N_1(x,y) \quad N_2(x,y) \quad N_3(x,y)]\begin{Bmatrix} \psi_1 \\ \psi_2 \\ \psi_3 \end{Bmatrix}$

C $\quad N_i = \frac{1}{2A}(a_i + b_i x + c_i y)$

C $\quad a_1 = x_2y_3 - x_3y_2 \qquad b_1 = y_2 - y_1 \qquad c_1 = x_3 - x_2$

C $\quad a_2 = x_3y_1 - x_1y_3 \qquad b_2 = y_3 - y_1 \qquad c_2 = x_1 - x_3$

C $\quad a_3 = x_1y_2 - x_2y_1 \qquad b_3 = y_1 - y_2 \qquad c_3 = x_2 - x_1$

```
      DO 201 IE=1,NELEM
      II=NNUM(IE,1)
      JJ=NNUM(IE,2)
      KK=NNUM(IE,3)
C     Combining nodal numbers into the one element.
      XG1=COORD(II,1)
```

```
      XG2=COORD(JJ,1)
      XG3=COORD(KK,1)
      YG1=COORD(II,2)
      YG2=COORD(JJ,2)
      YG3=COORD(KK,2)
C     Coordinates of node1, 2, 3 of each element
      AREA=0.5*(XG2*(YG3-YG1) + XG1*(YG2-YG3) +
     * XG3*(YG1-YG2))
      IF(AREA.LE.0.) WRITE(6,202) IE
C     To check nodal connection of each element.
202   FORMAT(/,' !!! ERROR !!! ELEMENT NO. ', I5,
     * ' HAS NEGATIVE OR ZERO AREA ', /,
     * ' --- CHECK F.E. MODEL FOR NODAL COORDINATES',
     * ' AND ELEMENT NODAL CONNECTIONS ---'        )
      IF(AREA.LE.0.) STOP
      B1=YG2-YG3
      B2=YG3-YG1
      B3=YG1-YG2
      C1=XG3-XG2
      C2=XG1-XG3
      C3=XG2-XG1
      DO 203 I=1,2
      DO 203 J=1,3
      B(I,J)=0.
C     To clear the values of bi, ci.
203   CONTINUE
      B(1,1)=B1
      B(1,2)=B2
      B(1,3)=B3
```

```
      B(2,1)=C1
      B(2,2)=C2
      B(2,3)=C3
      DO 204 I=1,2
      DO 205 J=1,3
      B(I,J)=B(I,J)/(2.*AREA)
```

C $\frac{\partial N_i}{\partial x} = \frac{b_i}{2A}$ for I = 1, $\frac{\partial N_i}{\partial y} = \frac{c_i}{2A}$ for I=2

```
      BT(J,I)=B(I,J)
```

C $\frac{\partial N_j}{\partial x} = \frac{b_{ji}}{2A}$ for I = 1, $\frac{\partial N_j}{\partial y} = \frac{c_j}{2A}$ for I=2

```
205   CONTINUE
204   CONTINUE
      DO 206 I=1,3
      DO 206 J=1,3
      AKC(I,J)=0.
      DO 207 K=1,2
      AKC(I,J)=AKC(I,J) + BT(I,K)*B(K,J)
```

C AKC $= \frac{1}{4A^2}\left(b_i b_j + c_i c_j\right)$

```
207   CONTINUE
      AKC(I,J)=PER(IE)*AREA*AKC(I,J)
```

C AKC $= K_{ij} = \frac{k}{4A}\left(b_i b_j + c_i c_j\right)$

```
206   CONTINUE
```

C $\underset{(3\times3)}{[K]_e}\ \underset{(3\times1)}{\{\psi\}_e} = \underset{(3\times1)}{\{F\}_e}$

```
      CALL MATRIX(IE,NNUM,AKC,SYSK,MXNODE,MXELE)
C     To call subroutine for creating the system of equations.
201   CONTINUE
```

```
      RETURN
      END
C----------------------------------------------------
      SUBROUTINE
MATRIX(IE,NNUM,AKC,SYSK,MXNODE,MXELE)
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION AKC(3,3)
      DIMENSION SYSK(MXNODE,MXNODE)
      INTEGER NNUM(MXELE,3)
      NNODE=3
      DO 301 IR=1,NNODE
      DO 302 IC=1,NNODE
      IROW=NNUM(IE,IR)
      ICOL=NNUM(IE,IC)
      SYSK(IROW,ICOL)=SYSK(IROW,ICOL)+AKC(IR,IC)
C     ∑(element equations) => [K]sys {ψ}sys = {F}sys
302   CONTINUE
301   CONTINUE
      RETURN
      END
C----------------------------------------------------
      SUBROUTINE BOUNDARY(NODE,COND,SL,SYSK,
     * SYSF,MXNODE)
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION SYSK(MXNODE,MXNODE),SYSF(MXNODE),
     * SL(MXNODE)
      INTEGER COND(MXNODE)
      DO 401 IEQ=1,NODE
      IF(COND(IEQ).EQ.0) GOTO 401
```

```
C       To check conditions of calculation
        DO 402 IR=1,NODE
        IF(IR.EQ.IEQ) GOTO 402
        SYSF(IR)=SYSF(IR) - SYSK(IR,IEQ)*SL(IEQ)
C       To eliminate the values of {F}sys after the boundary conditions were
C       applied in the system of equations.
        SYSK(IR,IEQ)=0.
C       The values of [K]sys in 'IR' row equaled zero except KIR,IEQ.
402     CONTINUE
        DO 403 IC=1,NODE
        SYSK(IEQ,IC)=0.
C       The values of [K]sys in 'IC' column equaled zero.
403     CONTINUE
        SYSK(IEQ,IEQ)=1.
C       KIQ,IEQ in [K]sys = 1
        SYSF(IEQ)=SL(IEQ)
C       FIEQ in {F}sys = 1
 401    CONTINUE
        END
C----------------------------------------------------
        SUBROUTINE GAUSS(NEQ, SYSK, SYSF, SL, MXNODE)
        IMPLICIT REAL*8 (A-H,O-Z)
        DIMENSION SYSK(MXNODE,MXNODE), SYSF(MXNODE),
      * SL(MXNODE)
        CALL SCAL(NEQ, SYSK, SYSF, MXNODE)
C       To call subroutine for dividing the equations in [K]sys by using the
C       maximum coefficient of each row.
        DO 501 IP=1,NEQ-1
        CALL PIVOT(NEQ, SYSK,SYSF, MXNODE, IP)
```

```
C       To call subroutine for rearranging the equations.
        DO 502 IE=IP+1,NEQ
        RATIO=SYSK(IE,IP)/SYSK(IP,IP)
        DO 503 IC=IP+1,NEQ
        SYSK(IE,IC)=SYSK(IE,IC)-RATIO*SYSK(IP,IC)
C       Forward elimination of [K]sys
503     CONTINUE
        SYSF(IE)=SYSF(IE)-RATIO*SYSF(IP)
C       Forward elimination of [F]sys
502     CONTINUE
        DO 504 IE=IP+1,NEQ
        SYSK(IE,IP)=0.
504     CONTINUE
501     CONTINUE
        SL(NEQ)=SYSF(NEQ)/SYSK(NEQ,NEQ)
C       Back substitution of gas streamlines at 'n' node.
        DO 505 IE=NEQ-1,1,-1
        SUM=0.
        DO 506 IC=IE+1,NEQ
        SUM=SUM+SYSK(IE,IC)*SL(IC)
C       Back substitution of gas streamlines except 'n' node.
506     CONTINUE
        SL(IE)=(SYSF(IE)-SUM)/SYSK(IE,IE)
505     CONTINUE
        RETURN
        END
C-----------------------------------------------------
        SUBROUTINE SCAL(NEQ, SYSK, SYSF, MXNODE)
        IMPLICIT REAL*8 (A-H,O-Z)
```

```
      DIMENSION SYSK(MXNODE,MXNODE),SYSF(MXNODE)
      DO 601 IE=1,NEQ
      BIG=ABS(SYSK(IE,1))
C     To set the first coefficient of [K]sys was maximum coefficient of each
C     row.
      DO 602 IC=2,NEQ
      AMAX=ABS(SYSK(IE,IC))
      IF(AMAX.GT.BIG) BIG=AMAX
C     To search the maximum coefficient of each row.
602   CONTINUE
      DO 603 IC=1,NEQ
      SYSK(IE,IC)=SYSK(IE,IC)/BIG
C     To divide the coefficient of [K]sys by the maximum coefficient of
C     each row.
603   CONTINUE
      SYSF(IE)=SYSF(IE)/BIG
C     To divide the coefficient of {F}sys by the maximum coefficient of
C     each row.
601   CONTINUE
      RETURN
      END
C-------------------------------------------------------
      SUBROUTINE PIVOT(NEQ, SYSK, SYSF, MXNODE, IP)
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION SYSK(MXNODE,MXNODE), SYSF(MXNODE)
      JP=IP
      BIG=ABS(SYSK(IP,IP))
      DO 701 I=IP+1,NEQ
      AMAX=ABS(SYSK(I,IP))
```

```
      IF(AMAX.GT.BIG) THEN
      BIG=AMAX
      JP=I
C     To search the maximum coefficient of [K]sys in each column.
      ENDIF
701   CONTINUE
      IF(JP.NE.IP) THEN
      DO 702 J=IP,NEQ
      DUMY=SYSK(JP,J)
      SYSK(JP,J)=SYSK(IP,J)
      SYSK(IP,J)=DUMY
702   CONTINUE
C     To convert the coefficient in [K]sys.
      DUMY=SYSF(JP)
      SYSF(JP)=SYSF(IP)
      SYSF(IP)=DUMY
C     To convert the coefficient in {F}sys.
      ENDIF
      RETURN
      END
```

CURRICULUM VITAE

Name: Kitti Booncharoen

Date of Birth: Dec 1, 1977

Nationality: Thai

University Education:

1995-1999 Bachelor Degree of Science in Chemical Engineering, Kasetsart University, Bangkok, Thailand.